// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/PoolLP.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

contract PoolLP is Ownable {
	struct UserInfo {
		uint amount;
		uint rewardDebt;
		uint timestamp;
	}

	IERC20 public immutable TOKEN;
	IERC20 public immutable LP;
	address public STAKING;

	uint private constant PRECISION = 1e12;
	uint public constant START_AMOUNT = 5_000_000 ether;
	uint public constant EPOCH_AMOUNT = 3_600_000 ether;
	uint public constant EPOCH_DURATION = 1 days;
	uint public constant MAX_REWARDS = 400_000_000 ether;
	uint public totalStaked;
	uint public totalRewards;
	uint public tokenPerSecond;
	uint public epoch;
	uint public epochEndTime;
	uint public lastRewardTime;
	uint public START_TIME;

	uint public accTokenPerShare;
	mapping(address => UserInfo) public userInfo;

	constructor(IERC20 token, IERC20 lp) {
		TOKEN = token;
		LP = lp;
	}

	function init(address staking) external onlyOwner {
		require(STAKING == address(0), "Already initialized");
		STAKING = staking;
	}

	function start() external {
		require(msg.sender == STAKING, "Only Staking");
		START_TIME = block.timestamp;
		lastRewardTime = block.timestamp;
		epochEndTime = block.timestamp + EPOCH_DURATION;
		tokenPerSecond = START_AMOUNT / EPOCH_DURATION;
	}

	function stake(uint amount) external {
		_updatePool();

		uint newAmount = userInfo[msg.sender].amount + amount;
		userInfo[msg.sender] = UserInfo({
			amount: newAmount,
			rewardDebt: newAmount * accTokenPerShare / PRECISION,
			timestamp: block.timestamp
		});

		totalStaked += amount;
		LP.transferFrom(msg.sender, address(this), amount);
	}
	
	function unstake(uint amount) external {
		_updatePool();

		uint newAmount = userInfo[msg.sender].amount - amount;
		userInfo[msg.sender] = UserInfo({
			amount: newAmount,
			rewardDebt: newAmount * accTokenPerShare / PRECISION,
			timestamp: block.timestamp
		});

		totalStaked -= amount;
		LP.transfer(msg.sender, amount);
	}

	function claim() external {
		_updatePool();

		UserInfo storage user = userInfo[msg.sender];
		require(block.timestamp - user.timestamp > 1 days, "Cannot claim yet");

		uint pending = user.amount * accTokenPerShare / PRECISION - user.rewardDebt;
		user.rewardDebt =  user.amount * accTokenPerShare / PRECISION;
		user.timestamp =  block.timestamp;
		TOKEN.transfer(msg.sender, pending);
	}

	function emergencyUnstake() external {
		uint amount = userInfo[msg.sender].amount;
		userInfo[msg.sender] = UserInfo(0, 0, 0);
		totalStaked -= amount;
		LP.transfer(msg.sender, amount);
	}

	function updatePool() external {
		_updatePool();
	}
	
	function recover() external onlyOwner {
		require(block.timestamp > START_TIME + 30 days);
		TOKEN.transfer(msg.sender, TOKEN.balanceOf(address(this)));
	}

	function getPending(address who) external view returns (uint) {
		UserInfo storage user = userInfo[who];
		
		uint accTokenPerShareCached = accTokenPerShare;
		uint stakedLP = totalStaked;

		if (stakedLP == 0) return 0;

		uint multiplier = block.timestamp - lastRewardTime;
		uint rewards = multiplier * tokenPerSecond;
		
		if (totalRewards + rewards > MAX_REWARDS) {
			rewards = MAX_REWARDS - totalRewards;
		}

		accTokenPerShareCached += rewards * PRECISION / stakedLP;

		return user.amount * accTokenPerShareCached / PRECISION - user.rewardDebt;
	}

	function _updatePool() internal {
		if (totalRewards == MAX_REWARDS) return;
	
		uint256 stakedLP = totalStaked;
		if (stakedLP == 0) {
			require(lastRewardTime != 0, "Not started yet");
			lastRewardTime = block.timestamp;
			return;
		}

		if (block.timestamp > epochEndTime) {
			_updateAcc(epochEndTime, stakedLP);

			epoch += 1;
			epochEndTime = block.timestamp + EPOCH_DURATION;
			tokenPerSecond = (START_AMOUNT + EPOCH_AMOUNT * epoch) / EPOCH_DURATION;
		}

		_updateAcc(block.timestamp, stakedLP);
	}

	function _updateAcc(uint timestamp, uint stakedLP) internal {
		uint multiplier = timestamp - lastRewardTime;
		uint rewards = multiplier * tokenPerSecond;
		
		if (totalRewards + rewards > MAX_REWARDS) {
			rewards = MAX_REWARDS - totalRewards;
			totalRewards = MAX_REWARDS;
		} else {
			totalRewards += rewards;
		}

		accTokenPerShare += rewards * PRECISION / stakedLP;
		lastRewardTime = timestamp;
	}
}